
    AshFeistelCipher

Encrypted integer IDs for Ash resources using Feistel cipher
Database Support: PostgreSQL only (requires AshPostgres data layer and PostgreSQL database)

Overview
Sequential IDs (1, 2, 3...) leak business information. This library provides a declarative DSL to configure Feistel cipher encryption in your Ash resources, transforming sequential integers into non-sequential, unpredictable values automatically via database triggers.
Key Benefits:
	Secure IDs without UUIDs: Hide sequential patterns while keeping efficient integer IDs with adjustable bit size per column
	Deterministic encryption: Same insertion order always produces same encrypted ID (consistent seed data in dev/staging environments, unlike UUIDs/random integers)
	Automatic encryption: Database triggers handle encryption transparently
	Collision-free: One-to-one mapping

Default profile: time_bits: 15, data_bits: 38
For detailed information about the Feistel cipher algorithm, how it works, security properties, and performance benchmarks, see the feistel_cipher library documentation.

This package currently depends on feistel_cipher 1.1.0.
Installation
Using igniter (Recommended)
mix igniter.install ash_feistel_cipher

You can customize the installation with the following options:
	--repo or -r: Specify an Ecto repo for FeistelCipher to use.
	--functions-prefix or -p: Specify the PostgreSQL schema where the FeistelCipher functions will be created, defaults to public.
	--functions-salt or -s: Specify the constant value used in the Feistel cipher algorithm. A random value is generated by default if not specified. Must be between 0 and 2^31-1.

⚠️ Security Note: A cryptographically random salt is generated by default for each project. This ensures that encryption patterns cannot be analyzed across different projects. Never use the same salt across multiple production projects.

Manual Installation
If you need more control over the installation process, you can install manually:
	Add ash_feistel_cipher to your list of dependencies in mix.exs:
def deps do
  [
    {:ash_feistel_cipher, "~> 1.1"}
  ]
end

	Fetch the dependencies:
mix deps.get


	Install FeistelCipher separately:
mix igniter.install feistel_cipher --repo MyApp.Repo

If you need an explicit dependency pin, use:
{:feistel_cipher, "1.1.0"}

	Add :ash_feistel_cipher to your formatter configuration in .formatter.exs:
[
  import_deps: [:ash_feistel_cipher]
]


Upgrading from v0.x
See UPGRADE.md for the project migration guide.
If you need upstream details, refer to feistel_cipher v1.1.0 UPGRADE.md.
Usage
Quick Start
Add AshFeistelCipher extension to your Ash resource and use the declarative DSL:
defmodule MyApp.Post do
  use Ash.Resource,
    data_layer: AshPostgres.DataLayer,
    extensions: [AshFeistelCipher]

  postgres do
    table "posts"
    repo MyApp.Repo
  end

  attributes do
    integer_sequence :seq
    encrypted_integer_primary_key :id, from: :seq
    
    attribute :title, :string, allow_nil?: false
  end
end
Generate the migration:
mix ash.codegen create_post

This creates a migration with database triggers that automatically encrypt seq into id.
Generated Migration Example:
defmodule MyApp.Repo.Migrations.CreatePost do
  use Ecto.Migration

  def up do
    create table(:posts) do
      add :seq, :bigserial, null: false
      add :id, :bigint, null: false, primary_key: true
      add :title, :string, null: false
    end

    # Automatically generates trigger for seq -> id encryption
    execute(
      FeistelCipher.up_for_trigger("public", "posts", "seq", "id",
        time_bits: 15,
        time_bucket: 86400,
        encrypt_time: false,
        data_bits: 38,
        key: 1_984_253_769,
        rounds: 16,
        functions_prefix: "public"
      )
    )
  end

  def down do
    execute(FeistelCipher.down_for_trigger("public", "posts", "seq", "id"))
    drop table(:posts)
  end
end
How it works:
When you create a record, the database trigger automatically encrypts the sequential seq value:
# Create a post - seq and id are auto-generated
post = MyApp.Post.create!(%{title: "Hello World"})
# => %MyApp.Post{seq: 1, id: 3_141_592_653, title: "Hello World"}

# The encrypted id is deterministic and collision-free
post2 = MyApp.Post.create!(%{title: "Second Post"})
# => %MyApp.Post{seq: 2, id: 2_718_281_828, title: "Second Post"}

# You can query by the encrypted id
MyApp.Post.get!(3_141_592_653)
# => %MyApp.Post{seq: 1, id: 3_141_592_653, title: "Hello World"}
Advanced Examples
Custom ID range with data_bits:
attributes do
  integer_sequence :seq
  encrypted_integer_primary_key :id,
    from: :seq,
    data_bits: 32  # ~4 billion IDs (default: 38 = ~275 billion)
end
Disable time prefix (backward compatible with v0.x):
attributes do
  integer_sequence :seq
  encrypted_integer_primary_key :id, from: :seq, time_bits: 0, data_bits: 52
end
Multiple encrypted columns from same source:
attributes do
  integer_sequence :seq
  encrypted_integer_primary_key :id, from: :seq
  encrypted_integer :referral_code, from: :seq, allow_nil?: false
  
  attribute :title, :string, allow_nil?: false
end

# Each column uses a different encryption key, generating unique values:
# => %MyApp.Post{seq: 1, id: 3_141_592_653, referral_code: 8_237_401_928, title: "Hello"}
Using any integer attribute with from (e.g. optional postal code):
attributes do
  attribute :postal_code, :integer, allow_nil?: true
  encrypted_integer :encrypted_postal_code, from: :postal_code, allow_nil?: true
end
Backfill existing rows when adding a new encrypted column:
attributes do
  integer_sequence :seq

  encrypted_integer :public_id,
    from: :seq,
    data_bits: 32,
    time_bits: 0,
    allow_nil?: true
end
This generates a migration that:
	adds the column with an internal sentinel default
	creates the trigger for new rows
	backfills existing rows using FeistelCipher.backfill_for_v1_column/5

DSL Reference
integer_sequence: Auto-incrementing bigserial column
integer_sequence :seq
encrypted_integer: Encrypted integer column
The base form for encrypted columns. Automatically sets writable?: false, generated?: true
encrypted_integer :id, from: :seq, primary_key?: true, allow_nil?: false, public?: true
encrypted_integer :referral_code, from: :seq
default: is not supported for encrypted_integer. Values are generated from from:, and encrypted_integer always uses an internal sentinel to avoid bigserial generation.
encrypted_integer_primary_key: Shorthand for encrypted primary keys
Convenience helper equivalent to encrypted_integer with primary_key?: true, allow_nil?: false, public?: true pre-set.
encrypted_integer_primary_key :id, from: :seq
encrypted_integer_primary_key :id, from: :seq, data_bits: 32

Common Options for Encrypted Columns:
Required:
	from: Integer attribute to encrypt (can be any integer attribute)

Optional (⚠️ Treat changes as explicit migrations):
	time_bits (default: 15): Time prefix bits for backup optimization. Set to 0 for no time prefix
	time_bucket (default: 86400): Time bucket size in seconds
	time_offset (default: 0): Time offset in seconds applied before bucket calculation	Formula: time_value = floor((epoch + time_offset) / time_bucket)
	Sign convention: positive values move the boundary earlier in local time; negative values move it later
	Example: time_bucket: 86400, time_offset: 21600 shifts daily boundary from 00:00 UTC to 18:00 UTC (03:00 KST)


	With defaults (time_bits: 15, time_bucket: 86400, encrypt_time: false), the time prefix wraps after about 89 years 9 months
	encrypt_time (default: false): Whether to encrypt the time prefix
	data_bits (default: 38): Data encryption bit size (must be even)
	key: Custom encryption key (auto-generated from table/column names if not provided)
	rounds (default: 16): Number of Feistel rounds (higher = more secure but slower)
	functions_prefix (default: "public"): PostgreSQL schema where feistel functions are installed
	backfill? (default: true): Backfill existing rows when generating a migration for a new encrypted column. Set backfill?: false to leave existing rows at the internal sentinel value until you handle them explicitly.

Important: 
	allow_nil? on encrypted column must match from attribute's nullability
	For primary keys, prefer encrypted_integer_primary_key for cleaner syntax

Why time_offset Is Needed
Without time_offset, daily time_bucket boundaries are anchored to UTC midnight. In local operations this can split one business day into two buckets at awkward local times (for example, evening in the Americas or early morning in Europe).
time_offset allows teams to keep the same bucket size (for example, one day) while moving the boundary to an operational cutover hour (for example, 03:00 local). This is especially useful when encrypt_time: true is enabled and continuity must be controlled by configuration, not by reading the encrypted prefix.
In this DSL, time_offset is added to epoch before bucketing. So time_offset: 21600 (not -21600) is the correct setting for a 03:00 KST daily boundary.
License
MIT


  

    Upgrading AshFeistelCipher

From v0.14.0 or v0.15.0 to v1.0.0
What changed
	bits option renamed to data_bits (default changed from 52 to 38)
	New DSL options: time_bits (default: 15), time_bucket (default: 86400), encrypt_time (default: false)
	Depends on feistel_cipher 1.1.0 (PG functions use _v1 suffix)
	Recommended default profile for new resources: time_bits: 15, data_bits: 38

Steps
	Update source code — replace bits: with time_bits: 0, data_bits: in your Ash resources:

# Before (v0.x)
encrypted_integer_primary_key :id, from: :seq, bits: 52

# After (v1.0)
encrypted_integer_primary_key :id, from: :seq, time_bits: 0, data_bits: 52
If bits was not specified (default was 52), add time_bits: 0, data_bits: 52 explicitly.
Note: The project won't compile until all bits: usages are replaced, because bits: is no longer a valid option in v1.0.

	Upgrade database — generate the migration for upgrading PostgreSQL functions:

mix ash_feistel_cipher.upgrade

This composes feistel_cipher.upgrade to generate an Ecto migration template. Edit the generated migration to fill in your functions_salt and trigger details. See feistel_cipher UPGRADE.md for details.
	Regenerate Ash migrations:

mix ash.codegen --name upgrade_feistel_triggers_to_v1

In the generated migration's up function, replace down_for_trigger (or down_for_v1_trigger) with force_down_for_legacy_trigger to drop legacy triggers. Also in the down function, replace up_for_trigger with up_for_legacy_trigger and bits: with time_bits: 0, data_bits:.

From v0.13.x or earlier to v1.0.0
The feistel_cipher dependency changed its cipher algorithm in v0.14.0 (HMAC-SHA256 hardening), so encryption results are different from v0.13.x and earlier. See feistel_cipher UPGRADE.md for details on compatibility.
The DSL upgrade steps are the same as above.


  

    Changelog

1.0.0
	Renamed bits to data_bits.
	Added time_bits, time_bucket, and encrypt_time options.
	Switched to feistel_cipher v1 PG function and trigger flow.
	Recommended new default profile: time_bits: 15, data_bits: 38.

1.1.0
	Added backfill? support for encrypted_integer and encrypted_integer_primary_key.
	Added migration backfill generation via FeistelCipher.backfill_for_v1_column/5.
	encrypted_integer now uses an internal sentinel default to avoid bigserial generation.
	User-provided default: is no longer supported for encrypted_integer.
	Generated custom statement names now include both from and to, which may cause migration and snapshot churn.



  

    Rules for working with AshFeistelCipher

Overview
AshFeistelCipher encrypts sequential integer IDs using Feistel cipher to prevent business information leakage. It handles encryption automatically via PostgreSQL database triggers.
Database Support: PostgreSQL only (requires AshPostgres data layer)
Default profile: time_bits: 15, data_bits: 38
Installation
Recommended using igniter:
mix igniter.install ash_feistel_cipher

Key options:
	--repo or -r: Specify Ecto repo
	--functions-prefix or -p: PostgreSQL schema for functions (default: public)
	--functions-salt or -s: Feistel cipher salt (default: randomly generated)

⚠️ Security Note: A unique salt is automatically generated per project. Never use the same salt across multiple production projects.
Basic Usage
Simple Primary Key Encryption
defmodule MyApp.Post do
  use Ash.Resource,
    data_layer: AshPostgres.DataLayer,
    extensions: [AshFeistelCipher]

  postgres do
    table "posts"
    repo MyApp.Repo
  end

  attributes do
    integer_sequence :seq
    encrypted_integer_primary_key :id, from: :seq
    
    attribute :title, :string, allow_nil?: false
  end
end
Generate migration:
mix ash.codegen create_post

Multiple Encrypted Columns
Create multiple encrypted columns from the same source (each uses different key):
attributes do
  integer_sequence :seq
  encrypted_integer_primary_key :id, from: :seq
  encrypted_integer :referral_code, from: :seq, allow_nil?: false
end
Optional Integer Encryption
Nullable integer attributes can also be encrypted:
attributes do
  attribute :postal_code, :integer, allow_nil?: true
  encrypted_integer :encrypted_postal_code, from: :postal_code, allow_nil?: true
end
DSL Reference
integer_sequence
Declares an auto-incrementing bigserial column:
integer_sequence :seq
encrypted_integer
Encrypted integer column (automatically sets writable?: false, generated?: true):
encrypted_integer :id, from: :seq, primary_key?: true
encrypted_integer :referral_code, from: :seq, key: 12345
encrypted_integer :public_id, from: :seq, allow_nil?: true
default: is not supported for encrypted_integer. Values are generated from from:, and encrypted_integer always uses an internal sentinel to avoid bigserial generation.
encrypted_integer_primary_key
Shorthand for primary keys (automatically sets primary_key?: true, allow_nil?: false, public?: true):
encrypted_integer_primary_key :id, from: :seq
encrypted_integer_primary_key :id, from: :seq, time_bits: 15, data_bits: 38
Configuration Options
Required
	from: Integer attribute to encrypt (required)

Optional
⚠️ Treat changes as explicit migrations:
	time_bits (default: 15): Time prefix bits. Set to 0 for v0.x-compatible behavior.
	time_bucket (default: 86400): Time bucket size in seconds.
	encrypt_time (default: false): Whether to encrypt the time prefix.
	data_bits (default: 38): Data encryption bit size. Must be even.
	key: Encryption key (auto-generated from table/column names if not provided)
	rounds (default: 16): Number of Feistel rounds (higher = more secure but slower)
	functions_prefix (default: "public"): PostgreSQL schema where feistel functions are installed
	backfill? (default: true): Backfill existing rows when adding a new encrypted column. Set backfill?: false to leave existing rows at the internal sentinel value until you handle them explicitly.

Custom Bit Size Example
encrypted_integer_primary_key :id, 
  from: :seq,
  time_bits: 15,
  data_bits: 40  # ~1 trillion ID range
Important Rules
Nullability Consistency
The encrypted column's allow_nil? must match the source attribute:
# CORRECT
attribute :postal_code, :integer, allow_nil?: true
encrypted_integer :encrypted_postal_code, from: :postal_code, allow_nil?: true

# WRONG - will fail verification
attribute :postal_code, :integer, allow_nil?: true
encrypted_integer :encrypted_postal_code, from: :postal_code, allow_nil?: false
Source Attribute Must Exist
The attribute specified in from must be defined:
# WRONG - :seq not defined
attributes do
  encrypted_integer_primary_key :id, from: :seq  # Error!
end

# CORRECT
attributes do
  integer_sequence :seq
  encrypted_integer_primary_key :id, from: :seq
end
Migration Required For Parameter Changes
These options should be treated as immutable in-place once records exist:
	time_bits
	time_bucket
	encrypt_time
	data_bits
	key
	rounds

Changing them requires data migration.
How It Works
Database triggers handle encryption automatically:
post = MyApp.Post.create!(%{title: "Hello"})
# => %MyApp.Post{seq: 1, id: 3_141_592_653, ...}

post2 = MyApp.Post.create!(%{title: "World"})
# => %MyApp.Post{seq: 2, id: 2_718_281_828, ...}
	Sequential seq → Non-sequential id via automatic encryption
	Deterministic (same seq always produces same id)
	Collision-free (one-to-one mapping)

Migration
mix ash.codegen automatically includes trigger creation code in migrations:
def up do
  create table(:posts) do
    add :seq, :bigserial, null: false
    add :id, :bigint, null: false, primary_key: true
  end

  execute(
    FeistelCipher.up_for_trigger("public", "posts", "seq", "id",
      time_bits: 15,
      data_bits: 38,
      key: 1_984_253_769,
      rounds: 16,
      functions_prefix: "public"
    )
  )
end

def down do
  execute(FeistelCipher.down_for_trigger("public", "posts", "seq", "id"))
  drop table(:posts)
end
Testing
Use standard Ash testing patterns:
test "encrypted IDs are generated" do
  post = MyApp.Domain.create_post!(%{title: "Test"})
  
  assert post.id != post.seq
  assert post.id > 0
  
  # Deterministic - same seq always produces same id
  post2 = MyApp.Domain.create_post!(%{title: "Test2"})
  assert post.id != post2.id
end
Common Pitfalls
Using UUID Instead
If you need UUIDs, use Ash's built-in uuid_primary_key:
# Use Feistel for integer IDs
encrypted_integer_primary_key :id, from: :seq

# Use UUID for random IDs
uuid_primary_key :id
Exposing Sequential IDs
Don't expose sequential IDs directly:
# BAD - exposes sequential pattern
%{id: post.seq}

# GOOD - use encrypted ID
%{id: post.id}
Changing Encryption Settings
Cannot change time_bits, time_bucket, encrypt_time, data_bits, key, or rounds after records are created. Requires data migration to change.
Upgrading to v1.1.0
	ash_feistel_cipher 1.1.0 and this guide now target feistel_cipher 1.1.0.
	Keep the default profile as time_bits: 15, data_bits: 38 for new resources.
	Replace legacy bits: N with time_bits: 0, data_bits: N in resource DSL.
	Run mix ash_feistel_cipher.upgrade to generate SQL function migration scaffolding.
	Run mix ash.codegen --name upgrade_feistel_triggers_to_v1 and apply trigger updates.
	For full upstream details, see feistel_cipher v1.1.0 UPGRADE.md.

        after?(_)

      


        Callback implementation for Spark.Dsl.Transformer.after?/1.



    


    
      
        after_compile?()

      


        Callback implementation for Spark.Dsl.Transformer.after_compile?/0.

Callback implementation for Spark.Dsl.Transformer.after?/1.

Callback implementation for Spark.Dsl.Transformer.after_compile?/0.

  


        

      


  

    
AshFeistelCipher.Verifier.AllowNilConsistency 
    



      
Verifies that when a from attribute allows nil, the encrypted attribute also allows nil.
This prevents potential errors where a nullable from value cannot be stored in a non-nullable encrypted attribute.

      




  

    
AshFeistelCipher.Verifier.MissingSource 
    



      
Verifies that all encrypted_integer attributes have a from configured.

      




  

    
mix ash_feistel_cipher.install 
    



      
Installs AshFeistelCipher and FeistelCipher
Example
mix igniter.install ash_feistel_cipher

Options
	--repo or -r — Specify an Ecto repo for FeistelCipher to use.
	--functions-prefix or -p — Specify the PostgreSQL schema prefix where the FeistelCipher functions will be created, defaults to public
	--functions-salt or -s — Specify the constant value used in the Feistel cipher algorithm. A random value is generated by default if not specified. Must be between 0 and 2^31-1.


      




  

    
mix ash_feistel_cipher.upgrade 
    



      
Generate a migration to upgrade FeistelCipher from v0.x to v1.0.0
Generates an Ecto migration that upgrades the PostgreSQL functions from v0.x to v1.0.0.
This composes feistel_cipher.upgrade with the same options.
Note: Before running this task, you must manually update your Ash resource DSL
to replace bits: with time_bits: 0, data_bits:. See UPGRADE.md for details.
Example
mix ash_feistel_cipher.upgrade

Options
	--repo or -r — Specify an Ecto repo for FeistelCipher to use.
	--functions-prefix or -p — Specify the PostgreSQL schema prefix (default: public)

